数学原論 8 (高山)　2005 年 7 月 5 日

8　差分化の復習

微分方程式の近似解を求める漸化式の導出方法 (差分法) については前々回のプリントで説明した. 今回はこの方法を別の角度から説明しよう.
今 $f(t)$ を未知関数とする微分方程式

$$f'(t) = a(t)f(t) + b(t), \quad f(0) = c$$

を解く問題を例として考える. ここで $a(t)$, $b(t)$ は時刻 t の関数であり, c は定数である. たとえば $a(t) = -1$, $b(t) = |\cos(t)|$, $c = 2$ とすれば上の微分方程式は $f'(t) = -f(t) + |\cos(t)|$, $f(0) = 2$ となる.
$f(t)$ は時刻 t おけるある量, たとえば温度とか細菌の数とか物体の位置や電流の大きさを表すとすれば, 初期条件 $f(0) = c$ での微分方程式を満たす関数 $f(t)$ を求めることは, 時刻 0 でのこれらの値から時刻 t での値を予想することにほかならない.
さて $h = \Delta t$ を微小な時間, たとえば $h = 0.00001$ とする. このとき微分の定義より $f'(t)$ は $(f(t+h) - f(t))/h$ にほぼ等しい. よって

$$f(t+h) \approx f(t) + hf'(t) = f(t) + (\Delta t)f'(t)$$

である. 微分方程式の右辺を用いて $f'(t)$ を書き換えると,

$$f(t+h) \approx f(t) + h(a(t)f(t) + b(t)) \tag{1}$$

となる. つまり $f(t)$ の $h = \Delta t$ 秒後の値 $f(t+h)$ は $f(t) + h(a(t)f(t) + b(t))$ に大体等しいということになる. つまり (1) は <u>微小時間 h 秒未来</u> の f の値を現在の f の値で表す式とみなせる.
(1) より $t = 0$ とすれば,

$$f(h) \approx f(0) + h(a(0)f(0) + b(0)) = c + h(a(0)c + b(0))$$

である. これで時刻 $t = h$ での $f(t)$ の近似値が求まった. 次に (1) で $t = h$ とすれば,

$$f(h+h) \approx f(h) + h(a(h)f(h) + b(h)),$$

$t = 2h$ とすれば

$$f(2h+h) \approx f(2h) + h(a(2h)f(2h) + b(2h))$$

となり, 時刻 $t = 2h$, $t = 3h$ の $f(t)$ の近似値が次々とさだまっていくこととなる. まとめると漸化式

$$f_{k+1} = f_k + h(a(hk)f_k + b(hk)), \quad f_0 = c \tag{2}$$

で数列 f_k を決めていけばそれが時刻 $t = hk$ での $f(t)$ の近似値となるのである.

9 連立方程式の差分化

問 9.1 上の考え方を用いて $y_1(t)$, $y_2(t)$ を未知関数とする次の連立の微分方程式の近似解を求めるプログラムを書きなさい．((f_k, g_k) の値をプロットしていくプログラムを書きなさい．)

$$y_1' = (2-y_2)y_1, \quad y_2' = (2y_1-3)y_2, \quad y_1(0)=4, y_2(0)=1.$$

参考: 種族 2 は種族 1 を食べる．$y_1(t)$ は時刻 t における種族 1 の数．$y_2(t)$ は時刻 t における種族 2 の数．と解釈する場合もある．

答え．h を微小な数とする．$y_1(t)=f(t)$, $y_2(t)=g(t)$ とおこう．このとき微分方程式より

$$\begin{aligned} f(t+h)-f(t) &\approx h(2-g(t))f(t) \\ g(t+h)-g(t) &\approx h(2f(t)-3)g(t) \end{aligned}$$

である．よって，

$$\begin{aligned} f(t+h) &\approx f(t)+h(2-g(t))f(t) \\ g(t+h) &\approx g(t)+h(2f(t)-3)g(t) \end{aligned}$$

この式の右辺は時刻 t での f, g の値できまる量であり，左辺は時刻 $t+h$ での f, g の値である．したがって漸化式

$$\begin{aligned} f_{k+1} &= f_k + h(2-g_k)f_k \\ g_{k+1} &= g_k + h(2f_k-3)g_k \\ f_0 &= 4, \\ g_0 &= 1 \end{aligned}$$

で数列 f_k, g_k をきめていけば f_k, g_k は 時刻 hk での $f(t)$, $g(t)$ の近似値となる．

```
100 SET WINDOW -1,10,-1,10
110 DRAW axes
120 LET  f=4
130 LET  g=1
140 LET  h=0.0001
150 FOR k=0 TO 200000
160    LET  f2 = f+h*(2-g)*f
170    LET  g2 = g+h*(2*f-3)*g
180    SET COLOR 0
190    PLOT POINTS: f,g
200    SET COLOR 4
210    PLOT POINTS: f2,g2
220    LET  f = f2
230    LET  g = g2
240 NEXT k
250 END
```

http://www.math.kobe-u.ac.jp/ ~ taka/2005/prayp.bas よりダウンロードできる．

問 9.2 初期条件 $f_0=4, g_0=1$ を変更して種族 1, 2 が絶滅したり絶滅寸前になる初期条件を実験的に決めよ．

10　カオス的力学系

解の関数が非常に複雑な形を描くような微分方程式があるということは 19 世紀から 20 世紀前半の数学者により理論的にある程度解明されていたが, その理論は難解で科学者一般の認識となることはなかった.
グラフィックを容易に表示できる計算機と差分法による微分方程式の解法がこの状況を一変させることとなる. 1970 年代以降のことである. 解の関数が初期値に敏感に依存し, 非常に複雑な形を描くような微分方程式は カオス的な微分方程式 と呼ばれている. 現在はカオス的な方程式を含むいわゆる 複雑系 の考え方が提唱され多くの科学者の研究対象となっている.
ここではカオス的な微分方程式の代表の一つである ローレンツ方程式 の解のグラフを描くプログラムを実行してその複雑な解を見てみよう.

課題 1

(1) 次のプログラムを入力して実行し, 解のグラフを観察せよ.
(2) 180, 190, 200 行で初期条件を設定している. これらの値を変えて解のグラフがどのように変わるか調べよ. 解のグラフは 8 の字を描く. 8 の下の部分を何度か回ってから 8 の上の部分へ移り上の部分を何度かまわる. これを繰り返す. 回る回数を数えよ. 動きが早すぎるときは 230 行目と 240 行目の間に

```
wait  delay  0.1
```

命令をいれるとよい. `0.1` は 0.1 秒停止することを意味する.

```
100 ! lorentz.bas. Solving p1’ = -a p1 + a p2,
110 ! p2’ = -p1 p3 + b p1 - p2
120 ! p3’ = p1 p2 + c p3
130 SET WINDOW -25,25,-25,25
140 DRAW axes
150 LET  a=10
160 LET  b=20
170 LET  c=2.66
180 LET  p1=0
190 LET  p2 = 3
200 LET  p3 = 0
210 LET  dt = 0.004
220 LET  t = 0
230 FOR t=0 TO 50 STEP dt
240    LET  q1 = p1+dt*(-a*p1+a*p2)
250    LET  q2 = p2+dt*(-p1*p3+b*p1-p2)
260    LET  q3 = p3+dt*(p1*p2-c*p3)
262    SET COLOR 0
264    PLOT POINTS: p1,p2
266    SET COLOR 4
270    PLOT POINTS: q1,q2
280    LET  p1 = q1
290    LET  p2 = q2
300    LET  p3 = q3
310 NEXT t
320 END
```

ローレンツ方程式は 3 つの未知関数 $p_1(t), p_2(t), p_3(t)$ についての次の連立微分方程式であ

る. a, b, c には適当な数字をいれる.

$$\begin{aligned} p_1' &= -ap_1 + ap_2, \\ p_2' &= -p_1p_3 + bp_1 - p_2, \\ p_3' &= p_1p_2 + cp_3 \end{aligned}$$

上のプログラムでは $(p_1(t), p_2(t))$ をプロットしている.
`http://www.math.kobe-u.ac.jp/ ~ taka/2004/lorentz2.txt` よりダウンロードできる.
差分化の解説 240 行目を変形すると

$$\frac{\mathtt{q1} - \mathtt{p1}}{\mathtt{dt}} = -\mathtt{a} * \mathtt{p1} + \mathtt{a} * \mathtt{p2}$$

である. `dt` を微小な数 (上のプログラムでは 0.004) としたとき, 変数 `q1` に $p_1(t + \mathtt{dt})$, 変数 `p1` に $p_1(t)$, 変数 `q2` に $p_2(t + \mathtt{dt})$, 変数 `p2` に $p_2(t)$, 変数 `q3` に $p_3(t + \mathtt{dt})$, 変数 `p3` に $p_3(t)$, が対応する. したがって 240 行目は

$$-\mathtt{a}p_1(t) + \mathtt{a}p_2(t) = \frac{p_1(t + \mathtt{dt}) - p_1(t)}{\mathtt{dt}} \simeq p_1'(t)$$

(一つ目の微分方程式) にほかならない. 同様に 250, 260 行目はそれぞれ 2 つ目, 3 つ目の微分方程式にほかならない.

11　2 階の微分方程式の差分化

たとえば, 単振動の方程式

$$\frac{d^2}{dt^2}y + y = 0$$

を数値的に解くことを考えてみよう.

物理既修者向け解説:
解く前に, この方程式の物理的意味を復習しておこう. 高校物理の最初の基本公式は

$$\text{質量} \times \text{加速度} = \text{力}$$

なる関係式である. この関係式を Newton の運動方程式という. 物体の時刻 t における位置を $q(t)$ とおくと, 速度は $q'(t)$, 加速度は $q''(t)$ である.
1 次元的に単振動をするバネについた質量 1 の物体 W を考えよう. 時刻 t における, W の位置を $y(t)$ とすることにする. ただし, バネが自然な長さにあるとき $y = 0$ とする. フックの法則によると, 自然な長さから y だけのびた (負のときはちぢんだとみなす) とき物体 W にかかる力は $-ky$ である. ここで k はバネできまる定数. ここで $k = 1$ と仮定して Newton の運動方程式を適用すると, 単振動の方程式

$$y'' + y = 0$$

を得る.

$y(0)=1$, $y'(0)=0$ を初期条件として $y(t)=\cos(t)$ がこの方程式の解であるが, この方程式を数値解法で解こう. 数値解法の利点は, cos や sin で解を書けないときでも, 微分方程式の近似解がわかることである. 式 (??) より, h が十分小さいとき,

$$\frac{y(t+h)-2y(t)+y(t-h)}{h^2}+y(t)$$

は大体 0 に等しい. したがって,

$$y(t+h)=2y(t)-y(t-h)-h^2y(t)$$

が近似的になりたつとしてよいであろう. この式は時刻 $t-h$ と t の y の値で, すこし先の時刻 $t+h$ の y の値を表す式である. また $y'(0)$ は $\frac{y(h)-y(0)}{h}$ にほぼ等しい. $y'(0)=0$ なので, $y(h)=y(0)$ が近似的に成り立つとしてよいであろう. したがって, 漸化式

$$y_{k+2}=2y_{k+1}-y_k-h^2h_{k+1},\ y_0=1, y_1=1 \tag{3}$$

を満たす数列を決めることにより, 解の近似をもとめることが可能であると予想できる. つまり y_k は hk 秒での y の値を近似していると予想される. このように解の近似を求める方法を 差分法 とよぶ. 漸化式 (3) を元の微分方程式の 差分化, または差分スキームとよぶ.

```
! 振動の方程式 y''+y = 0
LET  x1=1
LET  x2=1
LET  dt=0.01  ! これは十分小さい数なら何でもよい.
! y(k+2), y(k+1), y(k) が x3, x2, x1 に対応.
! h が dt に対応.
FOR t=0 TO 3 STEP dt
   LET x3 = 2*x2-x1-dt*dt*x2
   PRINT t,x1
   PLOT  LINES: t,x1;
   LET  x1=x2
   LET  x2=x3
NEXT t
END
```

上のプログラムで ! で始まる行は注釈行 (コメント行) であり, プログラムの実行には関係ない. ! から始まる注釈は行の途中から書きはじめてもよい. この場合は ! から行末までが注釈となる. 適宜注釈行を書くことにより, 人間がプログラムを読む助けとなる.
さて, これで微分方程式を近似的に解く問題が, 漸化式をみたす数列を求める問題になったのであるが, このような近似解が本当の解に収束するかとか, 全くことなる解しかとらえられない不安定現象が起こる場合があるとかの議論をやらないといけない. これはより上級の話題である.
注意: 漸化式は常に微小な数 h を含む. この h は十分小さければどんな数でも構わない. たとえば 0.01 とか 0.001 など. 近似解が本当の解に収束するという議論では, h を小さくしていけば差分法で求めた近似解が本当の解に収束して行く (どんどん近くなる) ということを証明する.

問 11.1

h が十分小さいとき次の近似公式が成り立つ.

$$\frac{dy}{dt}(T) \simeq \frac{y(T+h)-y(T)}{h}$$
$$\frac{d^2y}{dt^2}(T) \simeq \frac{y(T+h)-2y(T)+y(T-h)}{h^2}$$

この近似公式を用いて次の微分方程式を差分化しなさい. (微分方程式の解を近似する数列の漸化式を求めよ.) 数列の漸化式を用いて数列の値を決めて行くプログラムを書きなさい.

$$y''+3y=0, \quad y(0)=1, y'(0)=0.$$

さてこのあたりで読者には徹底的に復習をすることをすすめる. (1) 微分方程式 $y''+y=0$, $y(0)=1$, $y'(0)=0$ の差分化を独力で導けるだろうか? (2) 上のプログラムを自力で書けるだろうか? (1), (2) ができたら次へ進むとよいであろう. 孔子が論語の中でいっているように, “学びて時にこれを習う. またよろこばしからずや” である.

3. 第 3 回の講義では, 物体の落下を表す微分方程式 $y''=-9.8$ について学んだ. (a) 次のプログラムを入力, 実行せよ. どのようなグラフとなるか? (b) 物体が速度に比例する抵抗を受ける場合, 方程式は $y''=-9.8-ay'$ となる (a は適当な定数). これを解くプログラムに変更せよ.

```
100 ! 方程式 y''=-9.8, y(0)=100, y'(0)=0
110 SET WINDOW -1,5,-10,110
120 DRAW axes
130 LET  y1 = 100
140 LET  y2 = 100
150 LET  dt=0.001
160 ! y(k+1), y(k), y(k-1) が  y3, y2, y1 に対応. h が dt に対応.
170 FOR t=0 TO 5 STEP dt
180   LET  y3 = 2*y2 - y1 -9.8*dt*dt
190   PLOT LINES: t,y1
200   LET  y1 = y2
210   LET  y2 = y3
220 NEXT t
230 END
```

この問題も, 上と同じ方針で解ける. つまり,

$$y''(t)=-9.8-ay'(t)$$

を <u>とりあえず差分化してみればよい</u>. 差分化の計算を記す.

$$y''(t) \simeq \frac{y(t+h)-2y(t)+y(t-h)}{h^2}, \ y'(t) \simeq \frac{y(t+h)-y(t)}{h}, \ y''(t)=-9.8-ay'(t)$$

より

$$\frac{y(t+h)-2y(t)+y(t-h)}{h^2} \simeq -9.8-a\frac{y(t+h)-y(t)}{h}$$

両辺に h^2 を掛けて,

$$y(t+h) - 2y(t) + y(t-h) \simeq -9.8h^2 - ah\left[y(t+h) - y(t)\right]$$

移項すると

$$(1+ah)y(t+h) \simeq (2+ah)y(t) - y(t-h) - 9.8h^2$$

$y(t+h)$ を y3, y(t) をy2, $y(t-h)$ を y1 と置き換え, $\simeq$ を $=$ とおけば,

$$(1+a*h)*\mathtt{y3} = (2+a*h)*\mathtt{y2} - \mathtt{y1} - 9.8*h^2$$

したがって 180 行を次のように書き換えればよい.

```
180 LET y3 = ((2+a*dt)*y2- y1 - 9.8*dt*dt)/(1+a*dt)
```

12　Debug の仕方

12.1　print 文をはさむ

12.2　自分が計算機となったつもりになって下のメモリに格納されたデータを書き変えていく

漸化式

$$f_{k+2} = f_{k+1} + f_k, \quad f_1 = f_2 = 1$$

できまる数列 (フィボナッチ数列) の 100 項まで計算して表示するプログラム.

```
10 a=1
20 b=1
30 for k=1 to 100
40    print a
50    c = a+b
60    a = b
70    b = c
80 next k
```

アドレス	変数の名前	内容	内容	内容	内容	内容
	k	1	2			
	a	1	1			
	b	1	2			
	c	2	3			

1. “内容” は 50 行目の実行が終了した時の各変数の値.

問題: 上の図の空白をうめよ.
問題: 他のプログラムについても同様な図を作成してプログラムの実行を追いなさい.
参考:

1. アドレス (メモリの番地) は Basic のプログラムを読むときは必要ないが, C のプログラムを読むときは必須. 特にポインターを利用するとき.

2. C のプログラムを読むときは, 変数のサイズ, つまり 2 進数何桁の数を格納できるかにも注意.

`dim` を用いた実技試験問題の解答例. `jitugi1b.bas`
$V = LdI/dt$ (コイル), $I = CdV/dt$ (condenser), $V = RI$ (抵抗). LC 共振回路.